
Curvature combs and harmonized paths in
METAPOST

Linus Romer

Abstract

Most font editors offer curvature related tools. One
of these tools is the visualization of curvature via
curvature combs. Another tool is the so-called harmo-
nization, which makes the curvature continuous along
paths. An implementation of both tools in META-
POST will be presented. Curvature optimized paths
already play a significant role in METAFONT and
METAPOST and therefore some exemplary META-
POST paths will be examined for their curvature
behavior.

1 Curvature

The curvature in a point of a curve is the inverse
of the radius of the osculating circle at this point
(depicted here as a “curvature vector” on the opposite
side of the radius):

For straight segments, the curvature is constantly
zero, since the radius of the osculating circle is in-
finitely large. Vice versa, the curvature becomes
infinitely large when the radius of the osculating
circle tends to zero.

2 Curvature combs in METAPOST

We can assemble these curvature vectors into a cur-
vature comb:

The curvature may be additionally mapped to a color
and the gaps may be filled:

A figure as above can be achieved by

path p; p = ⟨path⟩ ; comb(p,300); draw p;

using the comb macro that will be presented here
(the 300 scales the curvature comb). We start by
defining the macro crossprod that returns the cross
product between two given vectors w⃗ and z⃗:

primarydef w crossprod z =

(xpart w * ypart z - ypart w * xpart z)

enddef;

Then the macro curv shall be applied to a path
p in order to return a “curvature vector” that is
orthogonal to the path in its initial point. The length
of the vector is proportional to the initial curvature
of the path:

vardef curv expr p =

save v,w,l; pair v,w;

v = direction 0 of p;

l = length v;

v := v/l;

w = (point 0 of p - 2*postcontrol 0 of p

+ precontrol 1 of p)/l;

2/3*(v crossprod w)/l*(v rotated -90)

enddef;

Here is the math behind this macro. A cubic Bézier
segment can be described by:(
x(t)

y(t)

)
= t3(3Q⃗− P⃗ + S⃗ − 3R⃗) + 3t2(P⃗ − 2Q⃗+ R⃗)

+ 3t(Q⃗− P⃗ ) + P⃗

P⃗

Q⃗ R⃗

S⃗t = 0 t = 1

The initial derivatives are then:(
ẋ(0)

ẏ(0)

)
= 3 (Q⃗− P⃗ )︸ ︷︷ ︸

=:v⃗

(
ẍ(0)

ÿ(0)

)
= 6 (P⃗ − 2Q⃗+ R⃗)︸ ︷︷ ︸

=:w⃗

The signed curvature is calculated by

)∣∣(ẋ

)∣∣3 .

Using l := |v⃗| we finally have the formula used in the
macro for the initial curvature:

3v⃗ × 6w⃗

(3l)3
=

2

3

v⃗ × w⃗

l3
=

2

3l
·
(
1

l
v⃗ × 1

l
w⃗

)
The divisions by l are necessary to prevent arith-
metic overflows. The special case

∣∣(ẋ(0)
ẏ(0)

)∣∣ = 0 is not

handled here. The curvature then would diverge to
±∞ (or be 0 if the cubic Bézier segment is a line
segment).

Now we define the curvature comb macro of a
path p by subdividing each segment in 50 parts and
filling an area over each part. Each part of the comb
is made of two subsequent “curvature” vectors k⃗, c⃗
that are scaled by a constant factor s given by the
user. The color depends on the average length of
them.

vardef comb(expr p,s) =

save q,c,k; path q; pair c,k;

for n = 0 upto length(p)-1:

c := s * curv subpath(n,n+1) of p;

for i = 1 upto 50:

k := c;

c := s * curv subpath(n+i/50,

n if i<25: +1 fi) of p;

q := subpath(n+(i-1)/50,n+i/50) of p;

fill q -- point 1 of q + c

-- point 0 of q + k

-- cycle withcolor

(1,1/(1+.1*.5[length c,length k]),0);

endfor

endfor

enddef;

The condition if i<25: +1 fi makes the subpath
as large as possible to get better accuracy.

A curvature of 0 is mapped to yellow and an
infinitely large curvature is mapped to red. This is
done by changing the green value between 1 and 0.
If you increase the .1, the green value tends faster
to 0.

3 Harmonize paths in METAPOST

In METAPOST, the code

z0 = (70,60); z1 = (0,30); z2 = (20,0);

draw z0{left} .. z1 .. z2{right};

produces the following curve:

z0

z1

z2

While the directions at the start and the end of the
path were set by the user, METAPOST has chosen
the angle of the path in z1 to equalize the so-called
mock curvature on both sides. The mock curvature is
a Taylor approximation of the real curvature (Hobby,
1986). After that, two cubic Bézier segments that
nearly minimize the curve energy have been drawn
between the given points.

z0

z1

z2

z0{left}

.. z1

.. z2{right}

The curvature comb in the preceding picture shows
that the curvature in z1 indeed is not continuous
but only near continuous. When a user sets the
direction in the joining knot, METAPOST has no
possibility to optimize the curvature in the joint and
the curvature often changes more abruptly in the
joint (see the following picture). However, this case
is frequent in type design because knots at horizontal
and vertical extrema are preferred over knots with
arbitrary direction.

z0

z1

z2

z0{left}

.. z1{down}

.. z2{right}

Fortunately, a METAPOST path can be modified to
a curvature continuous curve by the later defined
harmonize macro that moves the joining knot along
its tangent:

z0

z1

z2

harmonize z0{left}

.. z1{down}

.. z2{right}

4 The math of harmonization

Assume two adjoint cubic Bézier segments that have
the same direction in their joint and do not have
zero-handles. Furthermore, assume the joining knot
to not be a point of inflection. By translation and
rotation we can force one control point next to the
joining knot to lie on the origin of the coordinate
system and the joining knot tangent to lie on the
x-axis:

(..., d)

(0, 0)

(g, 0)

(i, 0)

(..., l)

We want to choose g such that the curvature is con-
tinuous. So the curvature on both sides of (g, 0) must

be equal:

2d

3g2
=

2l

3(i− g)2

In the special case d = l, we get g = i− g. Solving
for g we get

g =

{
d±

√
dl

d−l · i if d ̸= l,
i
2 else.

Since
√
dl is the geometric mean between d and l,

the solution d−
√
dl

d−l · i guarantees the joining knot to
lie between its control points.

(..., d)

(0, 0)

(d−
√
dl

d−l · i, 0)

(i, 0)

(..., l)

If either d or l is zero, g = d−
√
dl

d−l · i becomes either 0
or i. That means, that the joining knot will become
collocated with one of its control points, which gener-
ally should be avoided. One reason for this avoidance
is that the curvature might become infinitely large:

(..., 0)

(0, 0)

(i, 0)

(..., l)

undesireable

So, we will not alter the paths at all in the case of
either d or l being zero. This case occurs also when
a straight line goes over to a curve, which is quite
frequent in type design:

When the joining knot is a point of inflection, the
curvatures 2d

3g2 and 2l
3(i−g)2 must have different signs.

(..., d)

(0, 0)

(g, 0) (i, 0)

(..., l)

Hence, a curvature-continuous solution forces d =
l = 0. A curvature continuous solution then only has
to satisfy that all control points must lie on one line
e.g.:

In this situation, one could also satisfy further condi-
tions like the preservation of area. On the other hand,
having all four control points on the same line of the
two affected cubic Bézier segments is critical. Due
to rounding errors, such a conversion is likely to add
new points of inflection. So, instead of this, we will
only guarantee the absolute value of the curvature
to be continuous in the case of points of inflection
by moving the joining knot in the same manner as
before in between its control points:

Finally, the solution of setting

gnew =


gold if d = 0 or l = 0,
i
2 else if |d| = |l|,
|d|−

√
|dl|

|d|−|l| · i else

is chosen here and shall be the definition of harmo-
nization. We define a corresponding macro harmonize
that returns a harmonized version of a given path p.
In the generic case, the tangent in the joining knot is
not the x-axis (as depicted in the preceding figures),
so we calculate d and l as the height to the tangent
by cross products.

vardef harmonize expr p =

save t,u,d,l,n,q; pair t,u,q[];

n = length p;

for j = if cycle p: 0 else: 1 fi upto n-1:

q[j] = point j of p;

t := unitvector(direction j of p);

u := unitvector(point j of p

- precontrol j of p);

if eps > abs((u dotprod t) - 1): % smooth

l := abs(t crossprod

(precontrol j+1 of p - point j of p) );

d := abs(t crossprod

(postcontrol j-1 of p - point j of p) );

if not ( (l = 0) or (d = 0) ):

q[j] := if (d = l): .5 else:

((d-sqrt(d*l))/(d-l)) fi

[precontrol j of p,postcontrol j of p];

fi

fi

endfor

if not cycle p:

q[0] = point 0 of p;

q[n] = point n of p;

fi

q[0] % start returned path

for j = 0 upto n-1: % define new path

.. controls postcontrol j of p

and precontrol j+1 of p .. if (j = n-1)

and (cycle p): cycle else: q[j+1] fi

endfor

enddef;

A mostly equivalent algorithm has been published in
(Roach, 1990).

5 Examples of harmonization

Should you use harmonization? At least it does not
harm to consider it. Most of the time, the changes
are subtle as in the following bulb terminal:

original harmonized

And sometimes they are less subtle as in the following
calligraphic dots:

original harmonized

After harmonization, the dot has become more rounded
and may have lost its “personality”.

6 Smoothing out paths even more

Harmonization does not affect control points nor
the curvatures at other joining knots. Hence, it can
be easily used over several cubic Bézier segments.
Nonetheless, harmonized paths normally no longer
interpolate the knots they were originally meant to.
The author once thought it might therefore be a
good idea to leave the joining knots and move the
control points instead. Then we not only can make
the curvature continuous but also the the change of
curvature. The curve then becomes some kind of
“supersmooth”.

harmonized “supersmooth”

original

However, there are some problems that come with
this “supersmoothness”: It might introduce addi-
tional points of inflection (see below). Furthermore,
this will normally change the curvature at other knots
and break curvature continuity there.

harmonized “supersmooth”

original

